NAG Library Routine Document

G03BAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G03BAF computes orthogonal rotations for a matrix of loadings using a generalized orthomax criterion.

2 Specification

```
SUBROUTINE G03BAF (STAND, G, NVAR, K, FL, LDFL, FLR, R, LDR, ACC, MAXIT,        &
                   ITER, WK, IFAIL)

INTEGER            NVAR, K, LDFL, LDR, MAXIT, ITER, IFAIL
REAL (KIND=nag_wp) G, FL(LDFL,K), FLR(LDFL,K), R(LDR,K), ACC,                   &
                   WK(2*NVAR+K*K+5*(K-1))
CHARACTER(1)       STAND
```

3 Description

Let Λ be the p by k matrix of loadings from a variable-directed multivariate method, e.g., canonical variate analysis or factor analysis. This matrix represents the relationship between the original p variables and the k orthogonal linear combinations of these variables, the canonical variates or factors. The latter are only unique up to a rotation in the k-dimensional space they define. A rotation can then be found that simplifies the structure of the matrix of loadings, and hence the relationship between the original and the derived variables. That is, the elements, λ^*_{ij}, of the rotated matrix, Λ^*, are either relatively large or small. The rotations may be found by minimizing the criterion:

$$V = \sum_{j=1}^{k}\sum_{i=1}^{p}\left(\lambda^*_{ij}\right)^4 - \frac{\gamma}{p}\sum_{j=1}^{k}\left[\sum_{i=1}^{p}\left(\lambda^*_{ij}\right)^2\right]^2$$

where the constant γ gives a family of rotations with $\gamma = 1$ giving varimax rotations and $\gamma = 0$ giving quartimax rotations.

It is generally advised that factor loadings should be standardized, so that the sum of squared elements for each row is one, before computing the rotations.

The matrix of rotations, R, such that $\Lambda^* = \Lambda R$, is computed using first an algorithm based on that described by Cooley and Lohnes (1971), which involves the pairwise rotation of the factors. Then a final refinement is made using a method similar to that described by Lawley and Maxwell (1971), but instead of the eigenvalue decomposition, the algorithm has been adapted to incorporate a singular value decomposition.

4 References

Cooley W C and Lohnes P R (1971) *Multivariate Data Analysis* Wiley

Lawley D N and Maxwell A E (1971) *Factor Analysis as a Statistical Method* (2nd Edition) Butterworths

5 Parameters

1: STAND – CHARACTER(1) *Input*

On entry: indicates if the matrix of loadings is to be row standardized before rotation.

STAND = 'S'
The loadings are row standardized.

STAND = 'U'
The loadings are left unstandardized.

Constraint: STAND = 'S' or 'U'.

2: G – REAL (KIND=nag_wp) *Input*

On entry: γ, the criterion constant with $\gamma = 1.0$ giving varimax rotations and $\gamma = 0.0$ giving quartimax rotations.

Constraint: G $\geq$ 0.0.

3: NVAR – INTEGER *Input*

On entry: p, the number of original variables.

Constraint: NVAR $\geq$ K.

4: K – INTEGER *Input*

On entry: k, the number of derived variates or factors.

Constraint: K $\geq$ 2.

5: FL(LDFL,K) – REAL (KIND=nag_wp) array *Input/Output*

On entry: Λ, the matrix of loadings. FL(i,j) must contain the loading for the ith variable on the jth factor, for $i = 1,2,\ldots,p$ and $j = 1,2,\ldots,k$.

On exit: if STAND = 'S', the elements of FL are standardized so that the sum of squared elements for each row is 1.0 and then after the computation of the rotations are rescaled; this may lead to slight differences between the input and output values of FL.

If STAND = 'U', FL will be unchanged on exit.

6: LDFL – INTEGER *Input*

On entry: the first dimension of the arrays FL and FLR as declared in the (sub)program from which G03BAF is called.

Constraint: LDFL $\geq$ NVAR.

7: FLR(LDFL,K) – REAL (KIND=nag_wp) array *Output*

On exit: the rotated matrix of loadings, Λ^*. FLR(i,j) will contain the rotated loading for the ith variable on the jth factor, for $i = 1,2,\ldots,p$ and $j = 1,2,\ldots,k$.

8: R(LDR,K) – REAL (KIND=nag_wp) array *Output*

On exit: the matrix of rotations, R.

9: LDR – INTEGER *Input*

On entry: the first dimension of the array R as declared in the (sub)program from which G03BAF is called.

Constraint: LDR $\geq$ K.

10: ACC – REAL (KIND=nag_wp) *Input*

On entry: indicates the accuracy required. The iterative procedure of Cooley and Lohnes (1971) will be stopped and the final refinement computed when the change in V is less than ACC $\times$ max$(1.0, V)$. If ACC is greater than or equal to 0.0 but less than ***machine precision*** or if ACC is greater than 1.0, then ***machine precision*** will be used instead.

Suggested value: 0.00001.

Constraint: ACC $\geq$ 0.0.

11: MAXIT – INTEGER *Input*

On entry: the maximum number of iterations.

Constraint: MAXIT $\geq$ 1.

12: ITER – INTEGER *Output*

On exit: the number of iterations performed.

13: WK$(2 \times$ NVAR $+$ K $\times$ K $+ 5 \times$ (K $- 1$)) – REAL (KIND=nag_wp) array *Workspace*

14: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, K < 2,
or NVAR $<$ K,
or G < 0.0,
or LDFL $<$ NVAR,
or LDR $<$ K,
or ACC < 0.0,
or MAXIT ≤ 0,
or STAND $\neq$ 'S' or 'U'.

IFAIL $= 2$

The singular value decomposition has failed to converge. This is an unlikely error exit.

IFAIL $= 3$

The algorithm to find R has failed to reach the required accuracy in the given number of iterations. You should try increasing ACC or increasing MAXIT. The returned solution should be a reasonable approximation.

7 Accuracy

The accuracy is determined by the value of ACC.

8 Further Comments

If the results of a principal component analysis as carried out by G03AAF are to be rotated, the loadings as returned in the array P by G03AAF can be supplied via the parameter FL to G03BAF. The resulting rotation matrix can then be used to rotate the principal component scores as returned in the array V by G03AAF. The routine F06YAF (DGEMM) may be used for this matrix multiplication.

9 Example

This example is taken from page 75 of Lawley and Maxwell (1971). The results from a factor analysis of ten variables using three factors are input and rotated using varimax rotations without standardizing rows.

9.1 Program Text

```
    Program g03bafe

!     G03BAF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: g03baf, nag_wp, x04caf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: acc, g
      Integer                          :: i, ifail, iter, k, ldfl, ldr, lwk,   &
                                          maxit, nvar
      Character (1)                    :: stand
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: fl(:,:), flr(:,:), r(:,:), wk(:)
!     .. Executable Statements ..
      Write (nout,*) 'G03BAF Example Program Results'
      Write (nout,*)
      Flush (nout)

!     Skip heading in data file
      Read (nin,*)

!     Read in the problem size
      Read (nin,*) nvar, k, g, stand, acc, maxit

      ldfl = nvar
      ldr = k
      lwk = 2*nvar + k*k + 5*(k-1)
      Allocate (fl(ldfl,k),flr(ldfl,k),r(ldr,k),wk(lwk))

!     Read in loadings
      Read (nin,*)(fl(i,1:k),i=1,nvar)

!     Compute rotations
      ifail = 0
      Call g03baf(stand,g,nvar,k,fl,ldfl,flr,r,ldr,acc,maxit,iter,wk,ifail)

!     Display results
      ifail = 0
      Call x04caf('General',' ',nvar,k,flr,ldfl,'Rotated factor loadings', &
        ifail)
      Write (nout,*)
```

```
      Flush (nout)
      ifail = 0
      Call x04caf('General',' ',k,k,r,ldr,'Rotated matrix',ifail)

    End Program g03bafe
```

9.2 Program Data

```
G03BAF Example Program Data
  10 3 1.0 'U' 0.00001 20
 0.788 -0.152 -0.352
 0.874  0.381  0.041
 0.814 -0.043 -0.213
 0.798 -0.170 -0.204
 0.641  0.070 -0.042
 0.755 -0.298  0.067
 0.782 -0.221  0.028
 0.767 -0.091  0.358
 0.733 -0.384  0.229
 0.771 -0.101  0.071
```

9.3 Program Results

```
 G03BAF Example Program Results

 Rotated factor loadings
           1       2       3
  1   0.3293 -0.2888 -0.7590
  2   0.8488 -0.2735 -0.3397
  3   0.4500 -0.3266 -0.6330
  4   0.3450 -0.3965 -0.6566
  5   0.4526 -0.2758 -0.3696
  6   0.2628 -0.6154 -0.4642
  7   0.3322 -0.5614 -0.4854
  8   0.4725 -0.6841 -0.1832
  9   0.2088 -0.7537 -0.3543
 10   0.4229 -0.5135 -0.4089

 Rotated matrix
          1       2       3
 1   0.6335 -0.5337 -0.5603
 2   0.7580  0.5733  0.3109
 3   0.1553 -0.6217  0.7677
```